
AER_20071405_Tables_3_4_5_9.do

*README 
*______
*This do-file generates Tables 3, 4, 5 and 9 of the paper
*Command lines that are indented are of minor importance





*Program settings and variables
*______________________________

	clear
	set mem 100m
	log using AER_20071405_Tables_3_4_5_9, replace

use AER_20071405_dataset

	gen ATMwithdr=QL
	gen withdrvalue=VQ 
	gen numATMs=NLOC 
	gen numbranches=(b0+b2+b3+b4+b5+b6+b7+b8)/8 
	gen population=L 
	gen surface=1000*M
	gen enterprises=ent
	gen foreign=pctfor
	gen young=pctage1
	gen elderly=pctage3
	gen unemprate=u
	gen Flanders=FLA

	replace ATMwithdr=0 if ATMwithdr==.
	replace withdrvalue=0 if withdrvalue==.





*TABLE 3 SUMMARY STATISTICS
*__________________________

*all markets
*-----------
sum ATMwithdr withdrvalue [weight=population]
sum numATMs numbranches population surface enterprises foreign young elderly unemprate Flanders

*markets with at least one ATM
*-----------------------------
sum ATMwithdr withdrvalue [weight=population] if numATMs>0
sum numATMs numbranches population surface enterprises foreign young elderly unemprate Flanders if numATMs>0





*TABLE 4 ATM CASH WITHDRWAL DEMAND AND NUMBER OF ATMS
*____________________________________________________

	sort numATMs

*ATM withdr
*---------------
by numATMs: sum ATMwithdr [weight=population]
sum ATMwithdr [weight=population]

*withdr value
*----------------
by numATMs: sum withdrvalue [weight=population]
sum withdrvalue [weight=population]




*TABLE 5 REDUCED-FORM DEMAND REGRESSIONS
*_______________________________________

	gen logATMwithdr=ln(ATMwithdr)
	gen logwithdrvalue=ln(withdrvalue)
	gen lognumATMs=ln(numATMs)
	gen lognumbranches=ln(numbranches)
	
*ATM withdr
*---------------
reg logATMwithdr lognumATMs lognumbranches enterprises foreign young elderly unemprate Flanders

test lognumATMs==-lognumbranches



*withdr value
*----------------
reg logwithdrvalue lognumATMs lognumbranches enterprises foreign young elderly unemprate Flanders





*TABLE 9 SQUARE ROOT TRAVEL DISTANCE MEASURE VERSUS ACTUAL TRAVEL DISTANCE
*_________________________________________________________________________


	gen logsurface=ln(surface)
	
	gen dist=0.5*sqrt(surface/NLOC)
	gen lnd=ln(dist)
	gen lnad=ln(adist)

	gen dist0=0.5*sqrt(surface/b0)
	gen lnd0=ln(dist0)
	gen lnad0=ln(adist0)
	gen lnb0=ln(b0)

	foreach num of numlist 2/8 {
	gen dist`num'=0.5*sqrt(surface/b`num')
	gen lnd`num'=ln(dist`num')
	gen lnad`num'=ln(adist`num')
	gen lnb`num'=ln(b`num')
	}


*Table 9: average distances and average distance regression
*----------------------------------------------------------
*x=N
*---
sum dist adist
corr dist adist
reg lnad logsurface lognumATMs

*x=B1-B7
*-------
foreach num of numlist 2/8 {
sum dist`num' adist`num'
corr dist`num' adist`num'
reg lnad`num' logsurface lnb`num'
}

*x=B8
*----
sum dist0 adist0
corr dist0 adist0
reg lnad0 logsurface lnb0

*Note: 
*-----
*the constant fi0 in the average distance regressions in Table 9 is obtained by taking the exponent of the constant in the regressions reported here





	clear
	log close







AER_20070514_program.prg
new;screen on;cls;

"

STARTTIME
";
datestr(date);timestr(time);

"


INVESTMENT AND USAGE OF NEW TECHNOLOGIES: EVIDENCE FROM A SHARED ATM NETWORK";

"




CHOOSE MODEL OPTIONS IN SECTION 0.2 OF THE PROGRAM

To estimate the demand only model in Table 6 without (with) demographics set 
demandonly=1, dosimul=0, numsim=100, demo=0(1), actualdist=0, semilog=0, dens=0 (and costperkm=7.91/sqrt(1000)).

To estimate the demand and entry model in Table 7 without (with) demographics set 
demandonly=0, dosimul=0, numsim=100, demo=0(1), actualdist=0, semilog=0, dens=0 (and costperkm=7.91/sqrt(1000)).

To estimate the more general functional form average distance parameters in Table 9 and the demand and entry model in Table 10 set 
demandonly=0, dosimul=0, numsim=100, demo=1, actualdist=1, semilog=0, dens=0 (and costperkm=7.91/sqrt(1000)).

To estimate the demand and entry model with semilog demand specification in Table 10 set 
demandonly=0, dosimul=0, numsim=100, demo=1, actualdist=0, semilog=1, dens=0 (and costperkm=7.91/sqrt(1000)).

To estimate the demand and entry model with economies of density in Table 10 set 
demandonly=0, dosimul=0, numsim=100, demo=1, actualdist=0, semilog=0, dens=1 (and costperkm=7.91/sqrt(1000)).

To do the policy counterfactuals in Table 8 set 
demandonly=0, dosimul=1, numsim=100, demo=1, actualdist=0, semilog=0, dens=0 and costperkm=7.91/sqrt(1000).

To estimate the policy counterfactuals in Table A3 set 
demandonly=0, dosimul=1, numsim=100, demo=1, actualdist=0, semilog=0, dens=0 and costperkm=3.16/sqrt(1000).


";



/*MAIN PROGRAM*/





/*STEP 0. LOADING THE DATA, INCLUDING LIBRARIES, CHOOSING OPTIONS, CREATING VARIABLES*/


/*0.1 OPEN FILE AND INCLUDE LIBRARIES*/
open fi=aer_20071405_dataset.dat varindxi;

library maxlik,pgraph;
#include maxlik.ext;


/*0.2 CHOOSE MODEL OPTIONS*/
demandonly=0;  /*demandonly=1(0) means estimate demand only model (joint entry and demand model)*/
dosimul=0;  /*dosimul=1(0) means (do not) perform counterfactuals*/
numsim=100; /*sets the number of draws to perform the counterfactuals: to replicate counterfactuals in the paper, set equal to 100*/
demo=1;	/*demo=1(0) means that (no) demographics are included in total cash withdrawals and the cost ratio*/
actualdist=0; /*actualdist=0(1) means that the square root law (more general functional functional form) is used to compute average distances to nearest ATM or branch*/
semilog=0;  /*semilog=0(1) means that the logit (semilog) demand functional form is selected*/
dens=0; /*dens=1(0) means (no) economies of density case*/
costperkm=7.91/*3.16*//sqrt(1000); /*costperkm=3.16/sqrt(1000) means 0.1 euro per km, costperkm=7.91/sqrt(1000) means 0.25 euro per km*/

/*0.3  READ DATA*/
dt0=readr(fi,rowsf(fi));dt=dt0;nobs=rows(dt);


/*0.4 CREATE VARIABLES*/

/*DEMOGRAPHICS*/
L=dt[.,iL]; /*population*/
enterprises=dt[.,ient]; /*#enterprises at NIS code level in 100000*/
unemprate=dt[.,iu]; /*unemployment rate at NIS code level*/
M=dt[.,iM]; /*market surface in 1000 km²*/
foreign=dt[.,ipctfor]; /*% of foreigners*/
young=dt[.,ipctage1]; /*% of population in age group 1*/
elderly=dt[.,ipctage3]; /*% of population in age group 3*/
Flanders=dt[.,iFLA];  /*region dummies*/
constant=ones(nobs,1);
xb=constant~enterprises~foreign~young~elderly~unemprate~Flanders;   /*nobs x 7 matrix*/

/*ATMS*/
Nk=dt[.,iNLOC];   /*number of ATM locations*/
QL=dt[.,iQL]; /*per capita withdrawal volume*/

/*BRANCHES*/
bsmallshare=dt[.,ib1];   /*total number of branches of small banks=used for market share*/
bsmallprice=dt[.,ib0];   /*mean number of branches of small banks=used for price*/
b2=dt[.,ib2];b3=dt[.,ib3];b4=dt[.,ib4];b5=dt[.,ib5];b6=dt[.,ib6];b7=dt[.,ib7];b8=dt[.,ib8]; /*number of branches bank2-8=used for market share and price*/
bvec=bsmallprice~b2~b3~b4~b5~b6~b7~b8;
dpb1=dt[.,idpb1];dpb2=dt[.,idpb2];dpb3=dt[.,idpb3];dpb4=dt[.,idpb4];dpb5=dt[.,idpb5];dpb6=dt[.,idpb6];dpb7=dt[.,idpb7];dpb8=dt[.,idpb8];
/*deposits (in mio BEF) per branch: gives weight to branches*/
/*Note: the branch variables related to small banks (bsmallshare and bsmallprice) correspond to B8 in the paper, while the others (b2-b8) correspond to B1-B7 in the paper*/

/*BANK MARKET SHARE VARIABLES BASED ON APPROPRIATELY SCALED NUMBER OF BRANCHES*/
w=dpb1~dpb2~dpb3~dpb4~dpb5~dpb6~dpb7~dpb8;
bshare=bsmallshare~b2~b3~b4~b5~b6~b7~b8; /*nobs x 8 matrix*/
weightedshare=(w.*bshare)./sumc((w.*bshare)');  /*nobs x 8 matrix of which the rows sum to one*/

/*DISTANCES*/
/*average distance to nearest ATM*/
adist=missrv(dt[.,iadist],999); 
/*average distance to nearest branch of respective banks*/
adist1=missrv(dt[.,iadist0],999);adist2=missrv(dt[.,iadist2],999);adist3=missrv(dt[.,iadist3],999);adist4=missrv(dt[.,iadist4],999);
adist5=missrv(dt[.,iadist5],999);adist6=missrv(dt[.,iadist6],999);adist7=missrv(dt[.,iadist7],999);adist8=missrv(dt[.,iadist8],999);
bdist=adist1~adist2~adist3~adist4~adist5~adist6~adist7~adist8;

/*DISTANCE FUNCTIONAL FORM PARAMETERS*/
if actualdist .eq 0;
/*these are the square root law parameters*/
    Nscaler=0.5;NMpower=0.5;Npower=-0.5;
    Bscaler=0.5;BMpower=0.5;Bpower=-0.5;
else;
/*these are more general functional form parameters based on the actual distances and the regressions in Table 9*/
    { onam,om,ob,ostb,ovc,ostd,osig,ocx,orsq,oresid,odbw }=ols(0,ln(selif(adist,Nk .gt 0)),ln(selif(1000*M,Nk .gt 0))~ln(selif(Nk,Nk .gt 0)));
    Nscaler=exp(ob[1]);NMpower=ob[2];Npower=ob[3];
    BMpower={};Bpower={};Bscaler={};
    i=1;do while i .le cols(bdist);
        { onam,om,ob,ostb,ovc,ostd,osig,ocx,orsq,oresid,odbw }=ols(0,ln(selif(bdist[.,i],bvec[.,i] .gt 0)),ln(selif(1000*M,bvec[.,i] .gt 0))~ln(selif(bvec[.,i],bvec[.,i] .gt 0)));
        Bscaler=Bscaler~exp(ob[1]);BMpower=BMpower~ob[2];Bpower=Bpower~ob[3];
    i=i+1;endo;
endif;

/*VARIABLE TO AVOID MISSING VALUES IN THE MAXLIK PROCEDURES, E.G. DIVIDING BY ZERO, LOG OF ZERO*/
eps = machEpsilon;  /*smallest possible value*/




if demandonly==1;
/*STEP 1. DEMAND ONLY ESTIMATION*/

print "


DEMAND ONLY";

/*1.1 MAXLIK OPTIONS*/
maxset;
_max_CovPar=2;
_max_MaxIters=50000;
_max_Algorithm=5;


/*1.2 VARIABLE NAMES AND STARTING VALUES*/
let labelsdemand=alfa constant enterprises foreign young elderly unemprate Flanders sigma1 vBminvA;
demandparamstart=5.47|0.08|0.08|0.13|-1.07|3.97|-3.70|-0.33|0.36|-0.75;
if demo==0;
demandparamstart=6.18|0.25|zeros(6,1)|0.39|-0.59;
_max_Active=ones(2,1)|zeros(6,1)|ones(2,1);
endif;          
_max_ParNames=labelsdemand;

/*1.3 CALL PROCEDURE*/			
{demandparam,lldemandval,lldemandgrad,lldemandcovmat,lldemandret}=maxlik(constant,0,&lldemand,demandparamstart);
call maxprt(demandparam,lldemandval,lldemandgrad,lldemandcovmat,lldemandret);

/*1.4 RETRIEVE (OVERESTIMATED) LAMBDA*/
demandalfa=demandparam[1,1];
demandbeta=demandparam[rows(demandalfa)+1:rows(demandalfa)+cols(xb),1];
demandsigma1=demandparamstart[rows(demandalfa)+rows(demandbeta)+1,1];
demandvbminva=demandparam[rows(demandalfa)+rows(demandbeta)+rows(demandsigma1)+1,1];

demandvbminva;
demandalfa;


/*1.5 IMPLIED DEMAND PREDICTIONS*/
/*QBAR*/
{qbardemand}=qbarCR(meanc(xb)',demandbeta,0);
print "

QBARDEMAND"
qbardemand;

Aqbardemand=gradp(&qbarCRgrad,demandbeta);
Eqbardemand=missrv(lldemandcovmat[rows(demandalfa)+1:rows(demandalfa)+rows(demandbeta),rows(demandalfa)+1:rows(demandalfa)+rows(demandbeta)],0);
sterrorqbardemand=sqrt(Aqbardemand*Eqbardemand*Aqbardemand');

print "

STANDARD ERROR QBARDEMAND"
sterrorqbardemand;

/*sAi*/
{pA,pAplus,pAmin,pB}=prices(0,0,meanc(M),meanc(Nk),meanc(bvec)');
{sAidemand}=SAIPROC(pA,pB,demandvbminva|demandalfa);

/*sA*/
{sAdemand}=SAPROC(sAidemand,meanc(weightedshare)');
print "

SADEMAND"
sAdemand;

AsAdemand=gradp(&SAPROCgrad,demandvbminva|demandalfa);
EsAdemand=missrv((lldemandcovmat[rows(demandalfa)+rows(demandbeta)+rows(demandsigma1)+1,rows(demandalfa)+rows(demandbeta)+rows(demandsigma1)+1]~lldemandcovmat[rows(demandalfa)+rows(demandbeta)+rows(demandsigma1)+1,1])|(lldemandcovmat[1,rows(demandalfa)+rows(demandbeta)+rows(demandsigma1)+1]~lldemandcovmat[1,1]),0);
sterrorsAdemand=sqrt(AsAdemand*EsAdemand*AsAdemand');

print "

STANDARD ERROR SADEMAND"
sterrorsAdemand;

/*ELAST*/
{elastdemand}=ELASTPROC(demandalfa,pA,sAidemand,sAdemand,meanc(weightedshare)');
print "

ELASTDEMAND"
elastdemand;

Aelastdemand=gradp(&ELASTPROCgrad,demandvbminva|demandalfa);
Eelastdemand=missrv((lldemandcovmat[rows(demandalfa)+rows(demandbeta)+rows(demandsigma1)+1,rows(demandalfa)+rows(demandbeta)+rows(demandsigma1)+1]~lldemandcovmat[rows(demandalfa)+rows(demandbeta)+rows(demandsigma1)+1,1])|(lldemandcovmat[1,rows(demandalfa)+rows(demandbeta)+rows(demandsigma1)+1]~lldemandcovmat[1,1]),0);
sterrorelastdemand=sqrt(Aelastdemand*Eelastdemand*Aelastdemand');

print "

STANDARD ERROR ELASTDEMAND"
sterrorelastdemand;

/*PRICE ELAST*/
{priceelastdemand}=PRICEELASTPROC(demandalfa,pA,sAidemand,sAdemand,meanc(weightedshare)');
print "

PRICE ELASTDEMAND"
priceelastdemand;

Apriceelastdemand=gradp(&PRICEELASTPROCgrad,demandvbminva|demandalfa);
Epriceelastdemand=missrv((lldemandcovmat[rows(demandalfa)+rows(demandbeta)+rows(demandsigma1)+1,rows(demandalfa)+rows(demandbeta)+rows(demandsigma1)+1]~lldemandcovmat[rows(demandalfa)+rows(demandbeta)+rows(demandsigma1)+1,1])|(lldemandcovmat[1,rows(demandalfa)+rows(demandbeta)+rows(demandsigma1)+1]~lldemandcovmat[1,1]),0);
sterrorpriceelastdemand=sqrt(Apriceelastdemand*Epriceelastdemand*Apriceelastdemand');

print "

STANDARD ERROR PRICE ELASTDEMAND"
sterrorpriceelastdemand;




else;
/*STEP 2. JOINT ESTIMATION*/

print "


JOINT DEMAND AND ENTRY";


/*2.1 MAXLIK OPTIONS*/
maxset;
_max_CovPar=2;
_max_MaxIters=1000;
_max_Algorithm=5;


/*2.2 VARIABLE NAMES AND STARTING VALUES*/
let labelsjoint=cconstant centerprises cforeign cyoung celderly cunemprate cFlanders sigma2 alfa constant enterprises foreign young elderly unemprate Flanders sigma12 sigma1 vBminvA lambda;

if semilog==0;
    jointparamstart={10,0,0,0,0,0,0,0.5,2,0,0,0,0,0,0,0,0,0.5,0,0};
else;
    jointparamstart={10,0,0,0,0,0,0,0.5,1,0,0,0,0,0,0,0,0,0.5,1.17,0};
endif;
if dens==0;
    if semilog==0;
        _max_Active=ones(rows(jointparamstart)-1,1)|0;
    else;
        _max_Active=ones(rows(jointparamstart)-2,1)|0|0;
    endif;
endif;
if demo==0;
    if semilog==0;
        _max_Active=1|zeros(6,1)|ones(3,1)|zeros(6,1)|ones(3,1)|0;
    endif;
endif;
_max_ParNames=labelsjoint;


/*2.3 CALL PROCEDURE*/			
{jointparam,lljointval,lljointgrad,lljointcovmat,lljointret}=maxlik(constant,0,&lljoint,jointparamstart);
call maxprt(jointparam,lljointval,lljointgrad,lljointcovmat,lljointret);


/*2.4 RETRIEVE PARAMETER VALUES*/
/*KAPPA*/
kappa=jointparam[1:cols(xb),1];
/*VARCOVAR*/
sigma2=jointparam[rows(kappa)+1,1];
/*ALFA*/
alfa=jointparam[rows(kappa)+rows(sigma2)+1,1];
/*BETA*/
beta=jointparam[rows(kappa)+rows(sigma2)+rows(alfa)+1:rows(kappa)+rows(sigma2)+rows(alfa)+cols(xb),1];
/*VARCOVAR*/
sigma12=jointparam[rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+1,1];
sigma1=jointparam[rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+rows(sigma12)+1,1];
/*VPAR*/
vbminva=jointparam[rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+rows(sigma12)+rows(sigma1)+1,1];
/*LAMBDA*/
lambda=jointparam[rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+rows(sigma12)+rows(sigma1)+rows(vbminva)+1,1];


vbminva;
alfa;
lambda;



/*2.5 IMPLIED DEMAND AND COST PREDICTIONS*/
/*QBAR*/
{qbar}=qbarCR(meanc(xb)',beta,0);
print "

QBAR"
qbar;

Aqbar=gradp(&qbarCRgrad,beta);
Eqbar=missrv(lljointcovmat[rows(kappa)+rows(sigma2)+rows(alfa)+1:rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta),rows(kappa)+rows(sigma2)+rows(alfa)+1:rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)],0);
sterrorqbar=sqrt(Aqbar*Eqbar*Aqbar');

print "

STANDARD ERROR QBAR"
sterrorqbar;

/*sAi*/
{pA,pAplus,pAmin,pB}=prices(0,0,meanc(M),meanc(Nk),meanc(bvec)');
{sAi}=SAIPROC(pA,pB,vbminva|alfa);

/*sA*/
{sA}=SAPROC(sAi,meanc(weightedshare)');

print "

SA"
sA;

AsA=gradp(&SAPROCgrad,vbminva|alfa);
EsA=missrv(((lljointcovmat[rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+rows(sigma12)+rows(sigma1)+1,rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+rows(sigma12)+rows(sigma1)+1]~lljointcovmat[rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+rows(sigma12)+rows(sigma1)+1,rows(kappa)+rows(sigma2)+1])|(lljointcovmat[rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+rows(sigma12)+rows(sigma1)+1,rows(kappa)+rows(sigma2)+1]~lljointcovmat[rows(kappa)+rows(sigma2)+1,rows(kappa)+rows(sigma2)+1])),0);
sterrorsA=sqrt(AsA*EsA*AsA');

print "

STANDARD ERROR SA"
sterrorsA;

/*ELAST*/
{elast}=ELASTPROC(alfa,pA,sAi,sA,meanc(weightedshare)');
print "

ELAST"
elast;

Aelast=gradp(&ELASTPROCgrad,vbminva|alfa);
Eelast=missrv(((lljointcovmat[rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+rows(sigma12)+rows(sigma1)+1,rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+rows(sigma12)+rows(sigma1)+1]~lljointcovmat[rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+rows(sigma12)+rows(sigma1)+1,rows(kappa)+rows(sigma2)+1])|(lljointcovmat[rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+rows(sigma12)+rows(sigma1)+1,rows(kappa)+rows(sigma2)+1]~lljointcovmat[rows(kappa)+rows(sigma2)+1,rows(kappa)+rows(sigma2)+1])),0);
sterrorelast=sqrt(Aelast*Eelast*Aelast');

print "

STANDARD ERROR ELAST"
sterrorelast;

/*PRICE ELAST*/
{priceelast}=PRICEELASTPROC(alfa,pA,sAi,sA,meanc(weightedshare)');
print "

PRICE ELAST"
priceelast;

Apriceelast=gradp(&PRICEELASTPROCgrad,vbminva|alfa);
Epriceelast=missrv(((lljointcovmat[rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+rows(sigma12)+rows(sigma1)+1,rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+rows(sigma12)+rows(sigma1)+1]~lljointcovmat[rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+rows(sigma12)+rows(sigma1)+1,rows(kappa)+rows(sigma2)+1])|(lljointcovmat[rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+rows(sigma12)+rows(sigma1)+1,rows(kappa)+rows(sigma2)+1]~lljointcovmat[rows(kappa)+rows(sigma2)+1,rows(kappa)+rows(sigma2)+1])),0);
sterrorpriceelast=sqrt(Apriceelast*Epriceelast*Apriceelast');

print "

STANDARD ERROR PRICE ELAST"
sterrorpriceelast;

/*COST RATIO*/
if dens==0;
    {CR}=CRproc(meanc(xb)',1,kappa|lambda,0);
else;
    {CR}=CRproc(meanc(xb)',1,kappa|lambda,0);
    {CR2}=CRproc(meanc(xb)',2,kappa|lambda,0);
endif;

print "

FIXED COST OVER COST SAVINGS EVALUATED AT VARIABLE MEANS OVER ALL MARKETS"
CR;
if dens==1;
CR2;
endif;

A=gradp(&CRprocgrad,kappa|lambda);
if dens==1;
A2=gradp(&CRprocgrad2,kappa|lambda);
endif;

E=missrv((lljointcovmat[1:rows(kappa),1:rows(kappa)]~lljointcovmat[1:rows(kappa),rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+rows(sigma12)+rows(sigma1)+rows(vbminva)+1])|(lljointcovmat[rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+rows(sigma12)+rows(sigma1)+rows(vbminva)+1,1:rows(kappa)]~lljointcovmat[rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+rows(sigma12)+rows(sigma1)+rows(vbminva)+1,rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+rows(sigma12)+rows(sigma1)+rows(vbminva)+1]),0);

sterrorCR=sqrt(A*E*A');
if dens==1;
sterrorCR2=sqrt(A2*E*A2');
endif;


print "

STANDARD ERROR CR"
sterrorCR;
if dens==1;
sterrorCR2;
endif;

print "

CONFIDENCE BOUND CR"
CR-1.96*sterrorCR~CR+1.96*sterrorCR;
if dens==1;
CR2-1.96*sterrorCR2~CR2+1.96*sterrorCR2;
endif;


/*2.6 CORRELATIONS EPSILON1 AND EPSILON2, AND ETA1 AND ETA2 (WE ESTIMATED THE MODEL WITH EPSILON1=ETA1 AND EPSILON2=ETA2-ETA1*/
strucind=0;
"

CORRELATION EPSILON1 AND EPSILON2";
{correlationnonstructurals}=corrproc(sigma12|sigma1|sigma2);
correlationnonstructurals;

Acorrnon=gradp(&corrproc,sigma12|sigma1|sigma2);
Ecorrnon=missrv((lljointcovmat[rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+1,rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+1]~lljointcovmat[rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+1,rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+rows(sigma12)+1]~lljointcovmat[rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+1,rows(kappa)+1])|lljointcovmat[rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+rows(sigma12)+1,rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+1]~lljointcovmat[rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+rows(sigma12)+1,rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+rows(sigma12)+1]~lljointcovmat[rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+rows(sigma12)+1,rows(kappa)+1]|lljointcovmat[rows(kappa)+1,rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+1]~lljointcovmat[rows(kappa)+1,rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+rows(sigma12)+1]~lljointcovmat[rows(kappa)+1,rows(kappa)+1],0);
sterrorcorrnon=sqrt(Acorrnon*Ecorrnon*Acorrnon');

print "

STANDARD ERROR CORRELATION EPSILON1 AND EPSILON2"
sterrorcorrnon;

strucind=1;

"

CORRELATION ETA1 AND ETA2";
{correlationstructurals}=corrproc(sigma12|sigma1|sigma2);
correlationstructurals;

Acorr=gradp(&corrproc,sigma12|sigma1|sigma2);
Ecorr=missrv((lljointcovmat[rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+1,rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+1]~lljointcovmat[rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+1,rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+rows(sigma12)+1]~lljointcovmat[rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+1,rows(kappa)+1])|lljointcovmat[rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+rows(sigma12)+1,rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+1]~lljointcovmat[rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+rows(sigma12)+1,rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+rows(sigma12)+1]~lljointcovmat[rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+rows(sigma12)+1,rows(kappa)+1]|lljointcovmat[rows(kappa)+1,rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+1]~lljointcovmat[rows(kappa)+1,rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+rows(sigma12)+1]~lljointcovmat[rows(kappa)+1,rows(kappa)+1],0);
sterrorcorr=sqrt(Acorr*Ecorr*Acorr');

print "

STANDARD ERROR CORRELATION ETA1 AND ETA2"
sterrorcorr;


/*2.7 FIT*/

{covstructurals}=covproc(sigma12|sigma1|sigma2);
covmatstructurals=(sigma1^2~covstructurals)|(covstructurals~((sigma2^2)+(sigma1^2)+2*sigma12));
CholFact=chol(covmatstructurals);
fitnumsim=100;
fitNmaxsim=maxc(Nk)+10;
fitNsim=zeros(nobs,fitnumsim);

fitsim=1;do while fitsim .le fitnumsim;
seed=fitsim;
{rndQbar,rndCR}=draws(seed,nobs);
fitmin=zeros(nobs,fitNmaxsim);
fitNj=1;do while fitNj .le fitNmaxsim;
{fitQbar}=qbarCR(xb,beta,rndQbar);
{fitICR}=qbarCR(xb,-kappa,/**/-rndCR);
{fitpA,fitpAplus,fitpAmin,fitpB}=prices(0,0,M,fitNj,bvec);
{fitsAi}=SAIPROC(fitpA,fitpB,vbminva|alfa);
{fitsAiplus}=SAIPROC(fitpAplus,fitpB,vbminva|alfa);
{fitsAimin}=SAIPROC(fitpAmin,fitpB,vbminva|alfa);
{fitsA}=SAPROC(fitsAi,weightedshare);
{fitsAmin}=SAPROC(fitsAimin,weightedshare);
fitsAdifmin=(fitNj .ne 0).*fitsA-(fitNj .ne 0).*(fitNj .ne 1).*fitsAmin;
fitscalemin=((fitNj+(fitNj .eq 0))^(1+lambda))-((fitNj-1+(fitNj .eq 0))^(1+lambda));
fitmin[.,fitNj]=ln(fitQbar.*fitICR.*L)+ln(fitsAdifmin)-ln(fitscalemin);
fitNj=fitNj+1;endo;
fitmargvalplus=(fitmin .ge 0);
fitNsim[.,fitsim]=sumc(fitmargvalplus');
fitsim=fitsim+1;endo;

EN=meanc(fitNsim');

print "

CORRELATION BETWEEN PREDICTED and OBSERVED NUMBER OF ATM LOCATIONS";
corrx(EN~Nk);
print "

IMPLIED R²";
corrx(EN~Nk)^2;




if dosimul==1;
/*STEP 3. POLICY SIMULATIONS*/
if dens==0;
    Fx=2300;
else;
    Fx=2300/(1^lambda);
endif;
cA=0.1; /*value does not matter for profit differences*/
maxSub=round((Fx/100)+0.5)*100+200;
rowsgrid=((maxSub/50)+1)*201;

CholFact=chol(covmatstructurals);

simmarketcols=nobs;


sqsimmarketN=zeros(numsim,simmarketcols);
sqsimmarketSdy=zeros(numsim,simmarketcols);
sqsimmarkettBmintA=zeros(numsim,simmarketcols);
sqsimmarketcBmincA=zeros(numsim,simmarketcols);
sqsimmarketQbar=zeros(numsim,simmarketcols);
sqsimmarketsA=zeros(numsim,simmarketcols);
sqsimmarketQA=zeros(numsim,simmarketcols);
sqsimmarketpi=zeros(numsim,simmarketcols);
sqsimmarketCS=zeros(numsim,simmarketcols);
sqsimmarketSN=zeros(numsim,simmarketcols);
sqsimmarketwelf=zeros(numsim,simmarketcols);

scsimmarketN=zeros(numsim,simmarketcols);
scsimmarketSdy=zeros(numsim,simmarketcols);
scsimmarkettBmintA=zeros(numsim,simmarketcols);
scsimmarketcBmincA=zeros(numsim,simmarketcols);
scsimmarketQbar=zeros(numsim,simmarketcols);
scsimmarketsA=zeros(numsim,simmarketcols);
scsimmarketQA=zeros(numsim,simmarketcols);
scsimmarketpi=zeros(numsim,simmarketcols);
scsimmarketCS=zeros(numsim,simmarketcols);
scsimmarketSN=zeros(numsim,simmarketcols);
scsimmarketwelf=zeros(numsim,simmarketcols);

pcsimmarketN=zeros(numsim,simmarketcols);
pcsimmarketSdy=zeros(numsim,simmarketcols);
pcsimmarkettBmintA=zeros(numsim,simmarketcols);
pcsimmarketcBmincA=zeros(numsim,simmarketcols);
pcsimmarketQbar=zeros(numsim,simmarketcols);
pcsimmarketsA=zeros(numsim,simmarketcols);
pcsimmarketQA=zeros(numsim,simmarketcols);
pcsimmarketpi=zeros(numsim,simmarketcols);
pcsimmarketCS=zeros(numsim,simmarketcols);
pcsimmarketSN=zeros(numsim,simmarketcols);
pcsimmarketwelf=zeros(numsim,simmarketcols);

fbsimmarketN=zeros(numsim,simmarketcols);
fbsimmarketSdy=zeros(numsim,simmarketcols);
fbsimmarkettBmintA=zeros(numsim,simmarketcols);
fbsimmarketcBmincA=zeros(numsim,simmarketcols);
fbsimmarketQbar=zeros(numsim,simmarketcols);
fbsimmarketsA=zeros(numsim,simmarketcols);
fbsimmarketQA=zeros(numsim,simmarketcols);
fbsimmarketpi=zeros(numsim,simmarketcols);
fbsimmarketCS=zeros(numsim,simmarketcols);
fbsimmarketSN=zeros(numsim,simmarketcols);
fbsimmarketwelf=zeros(numsim,simmarketcols);

market=1;
do while market .le nobs;

sqsim=zeros(numsim,10);
scsim=zeros(numsim,10);
pcsim=zeros(numsim,10);
fbsim=zeros(numsim,10);

sim=1;
do while sim .le numsim;

seed=sim*market;
{rndQbar,rndCR}=draws(seed,1);

{grid}=GRIDPROC(xb[market,.],M[market,1],L[market,1],bvec[market,.],weightedshare[market,.]);

grid[.,7]=round(grid[.,7]*100000)/100000;
grid[.,9]=round(grid[.,9]*100000)/100000;

{ICRsim}=CRproc(xb[market,.],1,-kappa|-lambda,/**/-rndCR); /*error=0 IN EST,-rndCR IN SIM*/
costsaving=Fx.*ICRsim;

/*sc*/
scstap1=selif(grid,((grid[.,1] .eq 0).*(grid[.,2] .eq 0)) .eq 1);
scstap2=selif(scstap1,scstap1[.,9] .eq maxc(scstap1[.,9]));
    scstap2bis=selif(scstap2,abs(Fx.*(scstap2[.,4]^((round((lambda)*100000)/100000)))-scstap2[.,3]) .eq minc(abs(Fx.*(scstap2[.,4]^((round((lambda)*100000)/100000)))-scstap2[.,3])));
    /**/scstap3=selif(scstap2bis,scstap2bis[.,3] .eq maxc(scstap2bis[.,3]));

/*pc*/
pcstap1=selif(grid,(grid[.,3] .eq 0));
pcstap2=selif(pcstap1,pcstap1[.,9] .eq maxc(pcstap1[.,9]));
    pcstap3=selif(pcstap2,abs(costsaving-(pcstap2[.,2]-pcstap2[.,1])) .eq minc(abs(costsaving-(pcstap2[.,2]-pcstap2[.,1]))));

/*fb*/
fbstap1=selif(grid,grid[.,9] .eq maxc(grid[.,9]));
    fbstap2=selif(fbstap1,abs(Fx.*(fbstap1[.,4]^((round((lambda)*100000)/100000)))-fbstap1[.,3]) .eq minc(abs(Fx.*(fbstap1[.,4]^((round((lambda)*100000)/100000)))-fbstap1[.,3])));
    fbstap2bis=selif(fbstap2,abs(costsaving-(fbstap2[.,2]-fbstap2[.,1])) .eq minc(abs(costsaving-(fbstap2[.,2]-fbstap2[.,1]))));
    /**/fbstap3=selif(fbstap2bis,fbstap2bis[.,3] .eq maxc(fbstap2bis[.,3]));

sqsim[sim,.]=selif(grid,((grid[.,1] .eq 0).*(grid[.,2] .eq 0).*(grid[.,3] .eq 0)) .eq 1)~costsaving;
scsim[sim,.]=scstap3~costsaving;
pcsim[sim,.]=pcstap3~costsaving;
fbsim[sim,.]=fbstap3~costsaving;

sim=sim+1;
endo;

sqsimmarketN[.,market]=sqsim[.,4];
sqsimmarketSdy[.,market]=sqsim[.,3];
sqsimmarkettBmintA[.,market]=sqsim[.,2]-sqsim[.,1];
sqsimmarketcBmincA[.,market]=sqsim[.,10];
sqsimmarketQbar[.,market]=sqsim[.,5];
sqsimmarketsA[.,market]=sqsim[.,6];
sqsimmarketQA[.,market]=sqsim[.,5].*sqsim[.,6];
sqsimmarketpi[.,market]=sqsim[.,7];
sqsimmarketCS[.,market]=sqsim[.,8];
sqsimmarketSN[.,market]=sqsim[.,3].*sqsim[.,4];
sqsimmarketwelf[.,market]=sqsim[.,9];

scsimmarketN[.,market]=scsim[.,4];
scsimmarketSdy[.,market]=scsim[.,3];
scsimmarkettBmintA[.,market]=scsim[.,2]-scsim[.,1];
scsimmarketcBmincA[.,market]=scsim[.,10];
scsimmarketQbar[.,market]=scsim[.,5];
scsimmarketsA[.,market]=scsim[.,6];
scsimmarketQA[.,market]=scsim[.,5].*scsim[.,6];
scsimmarketpi[.,market]=scsim[.,7];
scsimmarketCS[.,market]=scsim[.,8];
scsimmarketSN[.,market]=scsim[.,3].*scsim[.,4];
scsimmarketwelf[.,market]=scsim[.,9];

pcsimmarketN[.,market]=pcsim[.,4];
pcsimmarketSdy[.,market]=pcsim[.,3];
pcsimmarkettBmintA[.,market]=pcsim[.,2]-pcsim[.,1];
pcsimmarketcBmincA[.,market]=pcsim[.,10];
pcsimmarketQbar[.,market]=pcsim[.,5];
pcsimmarketsA[.,market]=pcsim[.,6];
pcsimmarketQA[.,market]=pcsim[.,5].*pcsim[.,6];
pcsimmarketpi[.,market]=pcsim[.,7];
pcsimmarketCS[.,market]=pcsim[.,8];
pcsimmarketSN[.,market]=pcsim[.,3].*pcsim[.,4];
pcsimmarketwelf[.,market]=pcsim[.,9];

fbsimmarketN[.,market]=fbsim[.,4];
fbsimmarketSdy[.,market]=fbsim[.,3];
fbsimmarkettBmintA[.,market]=fbsim[.,2]-fbsim[.,1];
fbsimmarketcBmincA[.,market]=fbsim[.,10];
fbsimmarketQbar[.,market]=fbsim[.,5];
fbsimmarketsA[.,market]=fbsim[.,6];
fbsimmarketQA[.,market]=fbsim[.,5].*fbsim[.,6];
fbsimmarketpi[.,market]=fbsim[.,7];
fbsimmarketCS[.,market]=fbsim[.,8];
fbsimmarketSN[.,market]=fbsim[.,3].*fbsim[.,4];
fbsimmarketwelf[.,market]=fbsim[.,9];

market=market+1;
endo;


sqsimN=sumc(sqsimmarketN');
sqsimNisnul=sumc((sqsimmarketN .eq 0)');
sqsimtBmintA=meanc(sqsimmarkettBmintA');
sqsimcBmincA=meanc(sqsimmarketcBmincA');
sqsimQbar=sumc((sqsimmarketQbar.*(L./sumc(L))')');
sqsimsA=sumc((sqsimmarketsA.*(L./sumc(L))')');
sqsimQA=sumc((sqsimmarketQA.*(L./sumc(L))')');
sqsimpi=sumc(sqsimmarketpi');
sqsimCS=sumc(sqsimmarketCS');
sqsimSN=sumc(sqsimmarketSN');
sqsimwelf=sumc(sqsimmarketwelf');
sqsimSdy=sqsimSN./sqsimN;

scsimN=sumc(scsimmarketN');
scsimNisnul=sumc((scsimmarketN .eq 0)');
scsimtBmintA=meanc(scsimmarkettBmintA');
scsimcBmincA=meanc(scsimmarketcBmincA');
scsimQbar=sumc((scsimmarketQbar.*(L./sumc(L))')');
scsimsA=sumc((scsimmarketsA.*(L./sumc(L))')');
scsimQA=sumc((scsimmarketQA.*(L./sumc(L))')');
scsimpi=sumc(scsimmarketpi')-sqsimpi;
scsimCS=sumc(scsimmarketCS')-sqsimCS;
scsimSN=sumc(scsimmarketSN');
scsimwelf=sumc(scsimmarketwelf')-sqsimwelf;
scsimSdy=scsimSN./scsimN;

pcsimN=sumc(pcsimmarketN');
pcsimNisnul=sumc((pcsimmarketN .eq 0)');
pcsimtBmintA=meanc(pcsimmarkettBmintA');
pcsimcBmincA=meanc(pcsimmarketcBmincA');
pcsimQbar=sumc((pcsimmarketQbar.*(L./sumc(L))')');
pcsimsA=sumc((pcsimmarketsA.*(L./sumc(L))')');
pcsimQA=sumc((pcsimmarketQA.*(L./sumc(L))')');
pcsimpi=sumc(pcsimmarketpi')-sqsimpi;
pcsimCS=sumc(pcsimmarketCS')-sqsimCS;
pcsimSN=sumc(pcsimmarketSN');
pcsimwelf=sumc(pcsimmarketwelf')-sqsimwelf;
pcsimSdy=pcsimSN./pcsimN;

fbsimN=sumc(fbsimmarketN');
fbsimNisnul=sumc((fbsimmarketN .eq 0)');
fbsimtBmintA=meanc(fbsimmarkettBmintA');
fbsimcBmincA=meanc(fbsimmarketcBmincA');
fbsimQbar=sumc((fbsimmarketQbar.*(L./sumc(L))')');
fbsimsA=sumc((fbsimmarketsA.*(L./sumc(L))')');
fbsimQA=sumc((fbsimmarketQA.*(L./sumc(L))')');
fbsimpi=sumc(fbsimmarketpi')-sqsimpi;
fbsimCS=sumc(fbsimmarketCS')-sqsimCS;
fbsimSN=sumc(fbsimmarketSN');
fbsimwelf=sumc(fbsimmarketwelf')-sqsimwelf;
fbsimSdy=fbsimSN./fbsimN;



"

SIMULATION RESULTS

";

"
        N               N=0                  S              tB-tA           cB-cA             Qbar            sA              QA              deltapi         deltaCS            SN          deltawelf

";
mio=1000000;
meanc(sqsimN)~meanc(sqsimNisnul)~0~0~meanc(sqsimcBmincA)~meanc(sqsimQbar)~meanc(sqsimsA)~meanc(sqsimQA)~0~0~0~0;
meanc(scsimN)~meanc(scsimNisnul)~meanc(scsimSdy)~meanc(scsimtBmintA)~meanc(scsimcBmincA)~meanc(scsimQbar)~meanc(scsimsA)~meanc(scsimQA)~meanc(scsimpi/mio)~meanc(scsimCS/mio)~meanc(scsimSN/mio)~meanc(scsimwelf/mio);
meanc(pcsimN)~meanc(pcsimNisnul)~meanc(pcsimSdy)~meanc(pcsimtBmintA)~meanc(pcsimcBmincA)~meanc(pcsimQbar)~meanc(pcsimsA)~meanc(pcsimQA)~meanc(pcsimpi/mio)~meanc(pcsimCS/mio)~meanc(pcsimSN/mio)~meanc(pcsimwelf/mio);
meanc(fbsimN)~meanc(fbsimNisnul)~meanc(fbsimSdy)~meanc(fbsimtBmintA)~meanc(fbsimcBmincA)~meanc(fbsimQbar)~meanc(fbsimsA)~meanc(fbsimQA)~meanc(fbsimpi/mio)~meanc(fbsimCS/mio)~meanc(fbsimSN/mio)~meanc(fbsimwelf/mio);
"
";
stdc(sqsimN)~stdc(sqsimNisnul)~0~0~stdc(sqsimcBmincA)~stdc(sqsimQbar)~stdc(sqsimsA)~stdc(sqsimQA)~0~0~0~0;
stdc(scsimN)~stdc(scsimNisnul)~stdc(scsimSdy)~stdc(scsimtBmintA)~stdc(scsimcBmincA)~stdc(scsimQbar)~stdc(scsimsA)~stdc(scsimQA)~stdc(scsimpi/mio)~stdc(scsimCS/mio)~stdc(scsimSN/mio)~stdc(scsimwelf/mio);
stdc(pcsimN)~stdc(pcsimNisnul)~stdc(pcsimSdy)~stdc(pcsimtBmintA)~stdc(pcsimcBmincA)~stdc(pcsimQbar)~stdc(pcsimsA)~stdc(pcsimQA)~stdc(pcsimpi/mio)~stdc(pcsimCS/mio)~stdc(pcsimSN/mio)~stdc(pcsimwelf/mio);
stdc(fbsimN)~stdc(fbsimNisnul)~stdc(fbsimSdy)~stdc(fbsimtBmintA)~stdc(fbsimcBmincA)~stdc(fbsimQbar)~stdc(fbsimsA)~stdc(fbsimQA)~stdc(fbsimpi/mio)~stdc(fbsimCS/mio)~stdc(fbsimSN/mio)~stdc(fbsimwelf/mio);

endif;
"

ENDTIME
";
datestr(date);
timestr(time);


endif;
/*STEP 4. CLOSING THE MAIN PROGRAM*/
fi=close(fi);end;





/*LEVEL 1 PROCEDURES*/





/*1P1. DEMAND ONLY LOGL-PROCEDURE*/

/*1P1.1 CONFIGURATION AND LOCALS*/
proc lldemand(demandparamstart,x);
    local alfa,beta,sigma1,vbminva,p;


/*1P1.2 PARAMETERS*/
alfa=demandparamstart[1,1];
beta=demandparamstart[rows(alfa)+1:rows(alfa)+cols(xb),1];
sigma1=demandparamstart[rows(alfa)+rows(beta)+1,1];
vbminva=demandparamstart[rows(alfa)+rows(beta)+rows(sigma1)+1,1];


/*1P1.3 LIKELIHOOD FUNCTION*/
{p}=llfunction(zeros(cols(xb),1),0,alfa,beta,0,sigma1,vbminva,0,1);

    retp(ln(p));
endp;





/*1P4. JOINT DEMAND AND ENTRY LOGL-PROCEDURE*/

/*1P4.1 CONFIGURATION AND LOCALS*/
proc lljoint(jointparamstart,x);
    local kappa,sigma2,alfa,beta,sigma12,sigma1,vbminva,lambda,p;


/*1P4.2 PARAMETERS*/
kappa=jointparamstart[1:cols(xb),1];
sigma2=jointparamstart[rows(kappa)+1,1];
alfa=jointparamstart[rows(kappa)+rows(sigma2)+1,1];
beta=jointparamstart[rows(kappa)+rows(sigma2)+rows(alfa)+1:rows(kappa)+rows(sigma2)+rows(alfa)+cols(xb),1];
sigma12=jointparamstart[rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+1,1];
sigma1=jointparamstart[rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+rows(sigma12)+1,1];
vbminva=jointparamstart[rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+rows(sigma12)+rows(sigma1)+1,1];
lambda=jointparamstart[rows(kappa)+rows(sigma2)+rows(alfa)+rows(beta)+rows(sigma12)+rows(sigma1)+rows(vbminva)+1,1];

/*1P4.3 LIKELIHOOD FUNCTION*/
{p}=llfunction(kappa,sigma2,alfa,beta,sigma12,sigma1,vbminva,lambda,3);

    retp(ln(p));
endp;



/*1P5. ELAST*/
proc ELASTPROC(alfa,pA,sAi,sA,w);
if semilog==0;
    retp(-Npower*alfa*pA*(sumc(((sAi.*(1-sAi)).*w)')./sA));
else;
    retp(-Npower*alfa*pA);
endif;
endp;

proc ELASTPROCgrad(param);
local sAi,sA;
{sAi}=SAIPROC(pA,pB,param);
{sA}=SAPROC(sAi,meanc(weightedshare)');
if semilog==0;
    retp(-Npower*param[2,1]*pA*(sumc(((sAi.*(1-sAi)).*meanc(weightedshare)')')./sA));
else;
    retp(-Npower*param[2,1]*pA);
endif;
endp;




/*1P5bis. PRICE ELAST*/
proc PRICEELASTPROC(alfa,pA,sAi,sA,w);
if semilog==0;
    retp(-alfa*pA*(sumc(((sAi.*(1-sAi)).*w)')./sA));
else;
    retp(-alfa*pA);
endif;
endp;

proc PRICEELASTPROCgrad(param);
local sAi,sA;
{sAi}=SAIPROC(pA,pB,param);
{sA}=SAPROC(sAi,meanc(weightedshare)');
if semilog==0;
    retp(-param[2,1]*pA*(sumc(((sAi.*(1-sAi)).*meanc(weightedshare)')')./sA));
else;
    retp(-param[2,1]*pA);
endif;
endp;




/*1P6. DRAWS*/
proc (2) = draws(seed,nobs);
local draw;
rndseed seed;
draw=CholFact'rndn(2,nobs);
retp(draw[1,1:nobs]',draw[2,1:nobs]');
endp;




/*1P7. COVMATPROCS*/
proc covproc(param);
retp(param[1,1]+param[2,1]^2);
endp;

proc corrproc(param);
if strucind==0;
    retp((param[1,1])/(param[2,1]*param[3,1]));
elseif strucind==1;
    retp((param[1,1]+param[2,1]^2)/(param[2,1]*sqrt(((param[3,1]^2)+(param[2,1]^2)+2*param[1,1]))));
endif;
endp;



/*1P7. GRID CREATION*/
proc GRIDPROC(X,M,Size,Bchoice,w);
    local grid,pric,nr,Sdy,tA,tB,tAc,tBc,Nmaxsim,Nsim,Nj,margpi,margval,margvalplus,Qbar,sA,prof,CS,welf;

grid=zeros(rowsgrid,3+rows(X)*6);

pric=zeros(rowsgrid,3);

nr=1;
Sdy=0;
do while Sdy .le maxSub;

tA=0;
tB=-5;
do while tB .le 5;

if abs(tB) .lt 0.00001;
tB=0;
endif;

tBc=(tB .ge 0).*tB;
tAc=(tB .lt 0).*abs(tB); 

pric[nr,1:3]=tAc~tBc~Sdy;

nr=nr+1;
tB=tB+0.05;
endo;

Sdy=Sdy+50;
endo;

Nmaxsim=maxc(Nk)+10;
margval=zeros(rowsgrid,Nmaxsim);

Nj=1;
do while Nj .le Nmaxsim;

{margpi}=marginalprofitsandwelfare(pric[.,1],pric[.,2],pric[.,3],X,M,Size,Nj,Bchoice,w);	/*nobs x cols(Nsim)*/
margval[.,Nj]=margpi;

Nj=Nj+1;
endo;

margvalplus=(margval .ge 0);
Nsim=sumc(margvalplus');

{Qbar,sA,prof,CS,welf}=profitsandwelfare(pric[.,1],pric[.,2],pric[.,3],X,M,Size,Nsim,Bchoice,w);

grid=pric~Nsim~Qbar*ones(rows(grid),1)~sA~prof~CS~welf;

    retp(grid);
endp;





/*LEVEL 2 PROCEDURES*/





/*2P1. LLFUNCTION PROCEDURE*/

proc llfunction(kappa,sigma2,alfa,beta,sigma12,sigma1,vbminva,lambda,estcase);
    local Qbar,ICRest,pA,pAplus,pAmin,pB,sAi,sAiplus,sAimin,sA,sAplus,sAmin,sAdifplus,sAdifmin,Nchoicedummies,scaleplus,scalemin,plus,min,lnQlf,lnQL0sel,lnQLfsel,e,condplus,condmin,condsigma2,p1,p2,condp2,p3,p;

{Qbar}=qbarCR(xb,beta,0);
{ICRest}=qbarCR(xb,-kappa,0);
{pA,pAplus,pAmin,pB}=prices(0,0,M,Nk,bvec);

{sAi}=SAIPROC(pA,pB,vbminva|alfa);
{sAiplus}=SAIPROC(pAplus,pB,vbminva|alfa);
{sAimin}=SAIPROC(pAmin,pB,vbminva|alfa);

{sA}=SAPROC(sAi,weightedshare);
{sAplus}=SAPROC(sAiplus,weightedshare);
{sAmin}=SAPROC(sAimin,weightedshare);

sAdifplus=sAplus-(Nk .ne 0).*sA;
sAdifmin=(Nk .ne 0).*sA-(Nk .ne 0).*(Nk .ne 1).*sAmin;

scaleplus=((Nk+1)^(1+lambda))-((Nk)^(1+lambda));
scalemin=((Nk+(Nk .eq 0))^(1+lambda))-((Nk-1+(Nk .eq 0))^(1+lambda));

plus=ln(Qbar.*ICRest.*L)+ln(sAdifplus)-ln(scaleplus);
min=ln(Qbar.*ICRest.*L)+ln(sAdifmin)-ln(scalemin);

lnQLf=ln(Qbar)+ln(sA);

if estcase==1; /*demand only*/
lnQL0sel=selif(missrv(ln(QL),0),Nk .gt 0);
lnQLfsel=selif(lnQLf,Nk .gt 0);
e=lnQL0sel-lnQLfsel;
else;	/*joint demand and entry*/
e=missrv(ln(QL),0)-lnQLf;
endif;

if estcase ne 1;
    condplus=plus-(sigma12/(sigma1^2))*e;
    condmin=min-(sigma12/(sigma1^2))*e;
    condsigma2=sqrt((sigma2^2)-((sigma12^2)/(sigma1^2)));

    p1=(Nk .eq 0).*(constant-cdfn(plus/sigma2));
    p2=(Nk .ne 0).*(cdfn(min/sigma2)-cdfn(plus/sigma2));
    condp2=(Nk .ne 0).*(cdfn(condmin/condsigma2)-cdfn(condplus/condsigma2));
endif;
p3=(1/sigma1).*pdfn(e/sigma1);

if estcase==1;	/*demand only*/
p=p3;

else;	/*joint demand and entry*/
p=p1+condp2.*p3;
endif;

    retp(p);
endp;





/*2P2. MARGINAL PROFITS AND WELFARE PROCEDURE*/

proc marginalprofitsandwelfare(tA,tB,Sdy,X,M,Size,Nchoice,Bchoice,w);
    local Qbar,ICR,pA,pAplus,pAmin,pB,sAi,sAimin,sA,sAmin,sAdifmin,pimarg;

{Qbar}=qbarCR(X,beta,rndQbar); /*error=0 IN EST,rndqbar IN SIM*/
{ICR}=CRproc(X,1,-kappa|-lambda,/**/-rndCR); /*error=0 IN EST,-rndCR IN SIM*/
{pA,pAplus,pAmin,pB}=prices(tA,tB,M,Nchoice,Bchoice);
{sAi}=SAIPROC(pA,pB,vbminva|alfa); 
{sAimin}=SAIPROC(pAmin,pB,vbminva|alfa);
{sA}=SAPROC(sAi,w);
{sAmin}=SAPROC(sAimin,w);

sAdifmin=(Nchoice .ne 0).*sA-(Nchoice .ne 0).*(Nchoice .ne 1).*sAmin;

pimarg=(tA-tB+ICR.*Fx).*sAdifmin.*Qbar.*Size-(dens .ne 0)*(Fx.*(Nchoice.*(Nchoice^lambda)-(Nchoice-1).*(((Nchoice-1)+(Nchoice .eq 1)*eps)^lambda)))-(dens .eq 0)*Fx+Sdy;

    retp(pimarg);
endp;





/*2P3. PROFITS AND WELFARE PROCEDURE*/

proc (5) = profitsandwelfare(tA,tB,Sdy,X,M,Size,Nchoice,Bchoice,w);
    local Qbar,ICR,pA,pAplus,pAmin,pB,sAi,sA,prof,CS,welf;

{Qbar}=qbarCR(X,beta,rndQbar); /*error=0 IN EST,rndqbar IN SIM*/
{ICR}=CRproc(X,1,-kappa|-lambda,/**/-rndCR); /*error=0 IN EST,-rndCR IN SIM*/
{pA,pAplus,pAmin,pB}=prices(tA,tB,M,Nchoice,Bchoice);
{sAi}=SAIPROC(pA,pB,vbminva|alfa);
sAi=(Nchoice .ne 0).*sAi; 
{sA}=SAPROC(sAi,w);

/*ICR(1)=cs/F(1)=cs/(Fx*1) so cs=ICR(1)*Fx*/
prof=(tB-ICR.*Fx-cA).*Qbar.*Size+(tA-tB+ICR.*Fx).*sA.*Qbar.*Size-Nchoice.*(Fx.*((Nchoice+(Nchoice .eq 0)*eps)^lambda)-Sdy);
if semilog==0;
    CS=Qbar.*Size.*(((sumc((((Bchoice .eq 0).*((Nchoice .ne 0).*ln(exp(0-alfa*pA))+(Nchoice .eq 0).*ln(eps))+(Bchoice .ne 0).*((Nchoice .ne 0).*ln((exp(0-alfa*pA)+exp(vbminva-alfa*pB)))+(Nchoice .eq 0).*ln(exp(vbminva-alfa*pB)))).*w)'))/alfa));
else;
    CS=Qbar.*Size.*(((sumc((((Bchoice .ne 0).*((Nchoice .ne 0).*((vbminva-alfa*pB)+sAi)+(Nchoice .eq 0).*(vbminva-alfa*pB))).*w)'))/alfa)); /*V=VB*/
endif;
welf=prof+CS-Nchoice.*Sdy;

    retp(Qbar,sA,prof,CS,welf);
endp;





/*LEVEL 3 PROCEDURES*/





/*3P1. QBAR AND CR*/
proc qbarCR(demo,param,unobs);
    retp(exp(demo*param+unobs));
endp;

proc qbarCRgrad(param);
    retp(qbarCR(meanc(xb)',param,0));
endp;

proc CRproc(demo,Nchoice,param,unobs);
    local demopart;
{demopart}=qbarCR(demo,param[1:cols(xb),1],unobs);
retp(demopart.*((Nchoice)^(round(param[cols(xb)+1:rows(param),1]*100000)/100000)));
    /*retp(exp(-demo*param[1:cols(xb),1]+unobs).*((Nchoice)^(round(param[cols(xb)+1:rows(param),1]*100000)/100000)));*/
endp;

proc CRprocgrad(param);
    retp(CRproc(meanc(xb)',1,param,0));
endp;

proc CRprocgrad2(param);
    retp(CRproc(meanc(xb)',2,param,0));
endp;


/*3P2.sAi*/
proc SAIPROC(pA,pB,param);
if semilog==0;
    retp(1/(1+exp(param[1,1]-param[2,1]*(pB-pA)))); /*V=VB-VA*/
else;
    local sharestep;
    sharestep=exp(-param[1,1]+param[2,1]*(pB-pA));  /*V=VB-VA*/
    retp((sharestep.le 1).*sharestep+(sharestep.gt 1)); 
endif;
endp;


/*3P3.SA*/
proc SAPROC(sAi,w);
    retp(sumc((sAi.*w)'));
endp;

proc SAPROCgrad(param);
    local sAi;

{sAi}=SAIPROC(pA,pB,param);

    retp(sumc((sAi.*meanc(weightedshare)')'));
endp;


/*3P4. PRICES*/
proc (4) = prices(tA,tB,M,Nchoice,Bchoice);
    local pA,pAplus,pAmin,pB;

if actualdist .eq 0;
    pA=(Nchoice .eq 0)*(1+eps)+(Nchoice .ne 0).*(tA+costperkm*Nscaler*((1000*M)^NMpower).*((Nchoice+(Nchoice .eq 0)*(1+eps))^Npower));
    pAplus=(tA+costperkm*Nscaler*((1000*M)^NMpower).*((Nchoice+1)^Npower));
    pAmin=(Nchoice .eq 0)*(1+eps)+(Nchoice .eq 1)+(Nchoice .ne 0).*(Nchoice .ne 1).*(tA+costperkm*Nscaler*((1000*M)^NMpower).*(((Nchoice+(Nchoice .eq 0)*(1+eps)+(Nchoice .eq 1))-1)^Npower));

    pB=(Bchoice .eq 0)*(1+eps)+(Bchoice .ne 0).*(tB+costperkm*Bscaler.*((1000*M).^BMpower).*((Bchoice+(Bchoice .eq 0)*(1+eps)).^Bpower));
else;
    pA=(Nchoice .eq 0)*(1+eps)+(Nchoice .ne 0).*(tA+costperkm*Nscaler*((1000*M)^NMpower).*((Nchoice+(Nchoice .eq 0)*(1+eps))^Npower));
    pAplus=(tA+costperkm*Nscaler*((1000*M)^NMpower).*((Nchoice+1)^Npower));
    pAmin=(Nchoice .eq 0)*(1+eps)+(Nchoice .eq 1)+(Nchoice .ne 0).*(Nchoice .ne 1).*(tA+costperkm*Nscaler*((1000*M)^NMpower).*(((Nchoice+(Nchoice .eq 0)*(1+eps)+(Nchoice .eq 1))-1)^Npower));

    pB=(Bchoice .eq 0)*(1+eps)+(Bchoice .ne 0).*(tB+costperkm*Bscaler.*((1000*M.*ones(1,8)).^BMpower).*((Bchoice+(Bchoice .eq 0)*(1+eps)).^Bpower));
endif;

    retp(pA,pAplus,pAmin,pB);
endp;

File description 


 


This readme file describes how to replicate the tables in  


 


Investment and usage of new technologies: evidence from a shared ATM network 


 


By 


 


Ferrari Stijn, Verboven Frank and Degryse Hans 


 


Overview 


 


In order to replicate the tables in the paper, datasets and programs are provided for Stata (version 8) and Gauss (version 6.0). 


 


The variables contained in the datasets are described in the following Table: 


  


Variable Description 


FLA indicator variable for Dutch-speaking part of Belgium 


L population 


M surface (in 1,000 km²) 


NLOC number of shared ATMs 


QL monthly per capita number of cash withdrawals at shared ATMs 


VQ value per cash withdrawal at shared ATMs 


adist average distance to nearest ATM based on actual consumer and ATM locations 


adist0 average distance to nearest branch of small bank based on actual consumer and branch locations 


adistj (j=2…8) average distance to nearest branch of bank j (j=2…8) based on actual consumer and branch locations 


b0 average number of branches of small banks 


b1 total number of branches of small banks 


bj (j=2…8) total number of branches of bank j (j=2…8) 







dpb1 deposits (in mio BEF) per branch of small bank 


dpbj (j=2…8) deposits (in mio BEF) per branch of bank j (j=2…8) 


ent number of enterprises (divided by 100,000) 


pctage1 fraction of population under 18 


pctage3 fraction of population over 65 


pctfor fraction of foreigners in the population 


u unemployment rate 


 


Stata 


 


The dataset provided for Stata is AER_20071405_dataset.dta 


 


The related do-file is AER_20071405_Tables_3_4_5_9.do 


 


Running the do-file will open the dataset and generate Tables 3, 4, 5 and 9 of the paper. 


 


Gauss 


 


The dataset provided for Gauss consists of two files: aer_20071405_dataset.dat and aer_20071405.dht 


 


The related program is AER_20071405_program.prg 


 


Running the program will open the dataset and generate the tables for the different models in the paper, depending on the choice of the options 


selected by the user in Section 0.2 of the program. 


 


The following table provides the options to be selected in Section 0.2 of the program in order to replicate the tables for the different models in 


the paper and should be read horizontally: 


 


 







Option Model 


Description 


(Paper 


section) 


demandonly dosimul numsim demo actualdist semilog dens costperkm 


Replicates 


Table 


Demand only 


Benchmark 


(IV) 
1 0 100 0 or 1 0 0 0 


7.91 


/sqrt(1000) 
6 


Demand and entry 


Benchmark 


(IV) 
0 0 100 0 or 1 0 0 0 


7.91 


/sqrt(1000) 
7 


Actual 


distance 


(VI A) 


0 0 100 1 1 0 0 
7.91 


/sqrt(1000) 


9, 10 


(col 1) 


Semilog 


(VI B) 
0 0 100 1 0 1 0 


7.91 


/sqrt(1000) 


10 


(col 2) 


Economies of 


density 


(VI C) 


0 0 100 1 0 0 1 
7.91 


/sqrt(1000) 


10 


(col 3) 


Counterfactuals 


k=€0.25 


(V B) 
0 1 100 1 0 0 0 


7.91 


/sqrt(1000) 
8 


k=€0.10 


(Web 


Appendix B) 


0 1 100 1 0 0 0 
3.16 


/sqrt(1000) 
A3 


 






